Chapter 3
Measurement polynomials as controlled variables – Exercises

Johannes Jäschke and Sigurd Skogestad

Abstract In this chapter we present two exercises for finding controlled variables, which are polynomials in the measurements. Detailed solutions and maple source code are included so that the reader can easily follow the procedure.

3.1 Introduction

To illustrate concepts from described in the textbook, we present some small problem and go through the solution step by step. The reader is encouraged to experiment on his own to understand the ideas better.

For solving the CSTR case study, the `multires` package is required[1]

3.2 Simple excercise

Exercise 3.1 (Cox 1992). Check whether the two polynomials $f_1 = 2x^2 + 3x + 1$ and $f_2 = 7x^2 + x + 3$ have a common root in $\mathbb{C}$.

Solution 3.1. The resultant is the determinant of the Sylvester matrix:

Johannes Jäschke
Department of Chemical Engineering, NTNU Trondheim, Norway, e-mail: jaschke@chemeng.ntnu.no

Sigurd Skogestad
Department of Chemical Engineering e-mail: skoge@chemeng.ntnu.no

[1] The software can be downloaded at www-sop.inria.fr/galaad/software/multires

$$Res(f_1, f_2) = \det\left(\begin{bmatrix} 2 & 0 & 7 & 0 \\ 3 & 2 & 1 & 7 \\ 1 & 3 & 3 & 1 \\ 0 & 1 & 0 & 3 \end{bmatrix}\right) = 153 \neq 0 \tag{3.1}$$

There exist no common root since the resultant is nonzero.

Maple code for the Sylvester matrix example

```
with(LinearAlgebra):
f1  := 2*x^2+3*x+1;
f2  := 7*x^2+x+3;
Syl := SylvesterMatrix(f1,f2);
Res := Determinant(Syl);
```

3.3 Isothermal CSTR Case Study

Consider a CSTR as in Figure 3.1, with a feed stream containing component A and with two first order chemical reactions,

$$\begin{aligned} A &\longrightarrow B \quad r_1 = k_1 c_A \\ B &\longrightarrow C \quad r_2 = k_2 c_B. \end{aligned} \tag{3.2}$$

Of the products formed, B is the desired procuct, while C is an undesired side product. The manipulated variable is the feed stream q, which can be adjusted to achieve profitable performance. The operational objective is to maximize the concentration of the desired product.



Fig. 3.1 Isothermal CSTR

It is assumed that the unmeasured disturbances are k_1 and k_2, and all other variables are known except of c_B, which is assumed difficult to measure. The unmeasured variables are summarized in Table 3.1, and all measurements

Symbol	Description
k_1	Reaction constant for reaction 1
k_2	Reaction constant for reaction 2
c_B	Concentration of desired product

Table 3.1 Unmeasured variables

and known parameters are shown in Table 3.2. The task is to find a con-

Symbol	Description	Type	Value	Unit
q	Feed flow rate	Known input	varying	m^3/min
c_A	Outlet concentration A	Measurement	varying	kmol/m^3
c_C	Outlet concentration C	Measurement	varying	kmol/m^3
V	Tank volume	Known parameter	0.9	m^3
c_{AF}	Feed concentration A	Known parameter	10.0	kmol/m^3
c_{BF}	Feed concentration B	Known parameter	0.0	kmol/m^3
c_{CF}	Feed concentration C	Known parameter	0.0	kmol/m^3

Table 3.2 Known variables: Inputs, measurements and parameters

trolled variable which can be controlled using the total flow rate, and which maximizes the desired concentration.
Subtasks:

1. Set up the steady state component balances
2. Set up the optimization problem
3. Write the optimality conditions
4. Calculate the reduced gradient
5. Eliminate the unknown Variables

3.4 Solution

3.4.1 Component Balance

We do a The steady state component balances for the system read:

$$
\begin{aligned}
g_1 &= qc_{AF} - qc_A - k_1 c_A V = 0 \\
g_2 &= qc_{BF} - qc_B + k_1 c_A V - k_2 c_B V = 0 \\
g_2 &= qc_{CF} - qc_C + k_2 c_B V = 0
\end{aligned}
\tag{3.3}
$$

3.4.2 Optimization Problem

The objective function is

$$J = -c_B, \tag{3.4}$$

which we want to minimize subject to the process model:

$$\begin{aligned} \min J& \\ s.t.& \\ g_1 = 0& \\ g_2 = 0& \\ g_3 = 0& \end{aligned} \tag{3.5}$$

3.4.3 Optimality Conditions

We write the Lagrangian

$$\mathcal{L} = J(\boldsymbol{z}) + \lambda^{\mathrm{T}} g(\boldsymbol{z}), \tag{3.6}$$

where $\boldsymbol{z} = (c_A, c_B, c_C, q)^{\mathrm{T}}$. The first order optimality conditions are then

$$\begin{aligned} \nabla_{\boldsymbol{z}} J(\boldsymbol{z}) + \nabla_{\boldsymbol{z}} g(\boldsymbol{z}) &= 0 \\ g(\boldsymbol{x}) &= 0 \end{aligned} \tag{3.7}$$

Next we calculate the null-space of the constraints $\boldsymbol{N} = [n_1, n_2, n_3, n_4]^{\mathrm{T}}$ with

$$n_1 = (c_{AF} - c_A)/(q + k_1 V) \tag{3.8}$$

$$n_2 = -\frac{-qc_{BF} - c_{BF}k_1V + qc_B + c_Bk_1V - k_1Vc_{AF} + k_1c_AV}{(q + k_2V)(q + k_1V)} \tag{3.9}$$

$$\begin{aligned} n_3 = -\big(&-c_{CF}q^2 - c_{CF}qk_1V - c_{CF}k_2Vq - c_{CF}k_2V^2k_1 + c_Cq^2 + c_Cqk_1V \\ &+ c_Ck_2Vq + c_Ck_2V^2k_1 - k_2Vqc_{BF} - k_2V^2c_{BF}k_1 + k_2Vqc_B + k_2V^2c_Bk_1 \\ &- k_2V^2k_1c_{AF} + k_2V^2k_1c_A\big) \,/\, ((q + k_2V)(q + k_1V)q) \end{aligned} \tag{3.10}$$

$$n_4 = 1 \tag{3.11}$$

Eliminate Lagrangian multipliers using the null space, we write the optimality conditions:

$$\begin{aligned} c^v = \boldsymbol{N(z)}^{\mathrm{T}} \nabla_{\boldsymbol{z}} J(\boldsymbol{z}) &= 0 \\ g(\boldsymbol{z}) &= 0 \end{aligned} \tag{3.12}$$

Since we control c^v to zero, we need to consider only the numerator, which is

$$Num(c^v) := -qc_{BF} - c_{BF}k_1V + qc_B + c_Bk_1V - k_1Vc_{AF} + k_1c_AV; \quad (3.13)$$

This expression cannot be used for control yet, because it contains unknown variables. These have to be eliminated in the next step.

3.4.4 Eliminating Unknowns k_1, k_2 and c_B

We use the package `multires` to construct the matrix for the toric resultant

$$\boldsymbol{M} = \begin{bmatrix} -qcBF & q & -cBFV - VcAF + cAV & V & 0 & 0 & 0 \\ qcBF & -q & cAV & 0 & 0 & -V & 0 \\ qcAF - qcA & 0 & -cAV & 0 & 0 & 0 & 0 \\ 0 & 0 & qcBF & -q & cAV & 0 & -V \\ 0 & 0 & qcAF - qcA & 0 & -cAV & 0 & 0 \\ qcCF - qcC & 0 & 0 & 0 & 0 & V & 0 \\ 0 & 0 & qcCF - qcC & 0 & 0 & 0 & V \end{bmatrix} \quad (3.14)$$

3.4.5 The Determinant

The (factorized) determinant of $\boldsymbol{M}$ is

$$\tilde{c} = q^3c_AV^4(c_{AF}c_A + c_{AF}c_{CF} - c_{AF}c_C - c_A^2) \quad (3.15)$$

We see that the pre-factor $q^3c_AV^4$ is nonzero under operation, so the condition for optimal operation is only the last factor:

$$c = c_{AF}c_A + c_{AF}c_{CF} - c_{AF}c_C - c_A^2 = 0 \quad (3.16)$$

Now the we have ended up with a controlled variable combination which contains only known variables. It may be confirmed by the reader that controlling c to zero leads to the same solution as solving the optimization problem (3.5).

3.4.6 Maple Code

The maple code for the CSTR example is given below

Maple code for the CSTR example

```
#################################################################
## This is the file mapleCSTR.mpl
## Simple CSTR
## Johannes Jaeschke"
## Nov. 2010
################################################################

with(LinearAlgebra):with(VectorCalculus):
# Define the cost to minimize
J :=-cB;

# Setting up the constraints
g1  :=q*cAF - q*cA - k1*cA*V;
g2  :=q*cBF - q*cB + k1*cA*V - k2*cB*V;
g3  :=q*cCF - q*cC        + k2*cB*V;
g :=[g1,g2,g3];

# Derive to obtain first order optimality constraints
Jac := Jacobian(g,[cA,cB,cC,q]);
gradT := Jacobian([J],[cA,cB,cC,q]);

# Calculate the null space of the constraints
N := NullSpace(Jac): N := N[1];

# Transpose and mutlitpy
G := Transpose(gradT):
NT  := Transpose(N):
# The reduced gradient is then:
Cv  := VectorMatrixMultiply(NT,G):
cv  := simplify(numer(Cv[1])):  # pick only numerator

# Unknown variables to be eliminated
read("multires.mpl"):
varlist := [k2,k1,cB];   #unknown variables
polylist:= [cv,g1,g2,g3];  #optimality conditions
BigMat  := spresultant(polylist,varlist); #Construct resultant matrix
LargeMat:= det(BigMat);   # Calculate determinant
c := factor(LargeMat);  # factorize the CV
save Jac,grad,N,Cv,cv,c, "invariant"; # save results
```